
Writing nearly 20 years ago, Alan Manne (1995) highlighted the crucial role of the real annual
discount rate for cost benefit analysis in climate policy. Manne summarizes as follows:

With a real annual discount rate of 5% or more - and there are no significant climate
impacts for half a century - these impacts have a present value that is virtually negligible.
Within a cost-benefit framework, it then becomes exceedingly difficult to justify any
near-term actions other than no-regrets policies. In the greenhouse debate, it is important
to draw a clear distinction between prescriptive and descriptive reasoning. A philosopher
or an economist may counsel a low or a zero rate of time preference, but this advice does
not provide a good description of the collective outcome of individual choices. In
particular, it implies an unrealistically rapid increase in the rate of savings and investment.

Climate policy analysts who counsel a low rate of time preference face a fundamental challenge:
economic growth models with low or zero discount rates are inconsistent with market interest rates. In
the Ramsey optimal growth framework, changes in the utility time preference produce implausible
changes in aggregate investment without a significant change in the marginal productivity of capital.
Using technical language, Manne declares that integrated assessment models based on very low
discount rates “fail to pass the laugh test.”

Our paper picks up where Manne left off. We contemplate a framework for climate policy design
which treats the economy as a second best situation, adding ideas from Frank Ramsey’s 1927 paper, “A
Contribution to the Theory of Taxation”. To clarify our persepctive we begin from optimal taxation in
which we design and implement taxes to maximize social welfare. Commonly, the optimal taxation
involves minimizing the distortions caused by taxes which are levied to pay for public goods. In the
optimal tax framework with a limited set of tax instruments, distortion and inefficiency may be
unavoidable.

Some more words about the literature are needed here. Writing in the same volume where
Manne’s paper appeared, Schelling [1995] casts the climate problem in terms of integenerational
equity. Whether economics is able to address the deeper ethical issues is unclear. Rutherford [2013]
echos Schelling’s observations regarding the role of kinship in attitudes toward the welfare of
generations in the distant future. Hampicke [2011] likewise cautions about the limitations of the
discounted utilitarianism as a framework for the assessment of climate policy. Indeed, Pindyck [2013]

reminds us of a wide range of shortcomings in the set of integrated assessment models applied to
climate policy questions.

The key points we make in this paper are as follows:

i. It is possible to distingush prescriptive and descriptive elements in integrated assessment models
and thereby provide a plausible, logically consistent assessment of optimal policies under low or
zero discount rates.

ii. The introduction of separate private and social discount rates permits calibration to an exogenous
baseline growth path. This provides finer control over the assumptions underlying a given
simulation and clarifies the distinction between exogenous and endogenous variables.

iii. Bilevel programming imposes only limited implementational overhead. Whether the underlying
model is specified as an equilibrium system or as an optimization problem facing a representative
agent, modern mathematical programming tools only require that the modeller specify the social
welfare maximand.

iv. Computational experiments demonstrate that large scale model instances, even models which are
are non-convex in terms of the policy instruments, can be processed using standard extended
mathematical programming (EMP) solvers.

1 Dynamic General Equilibrium Analysis

Dynamic general equilibrium models are widely employed analytical frameworks for applied policy
analysis. These models are attractive largely because they can describe transitional impacts of policy
interventions. For example, carbon abatement measures planned

for implementation several years hence will alter the profitability of investments in coal-fired
generating capacity today, and the analyst requires models which can assess the implications of these
policies for current investment and employment. Faced with relatively few parsimonious alternatives,
the economists traditionally rely on an assumption of intertemporal optimizing behavior as a starting
point for performing such calculations.

The academic domain inhabited by dynamic economic equilibrium models can be
uncomfortable due to the heterogeneity of interested parties. Policy models may be evaluated either on
the basis of theoretical clarity or on the basis of empirical consistency. Simpler models are typically
more compelling when they are relatively transparent, but simpler formulations can be difficult to
reconcile with the demands of the policy analysis community.

This typically implies that a model should be able to provide congruence with a variety of
alternative competing forecasts of GDP growth, energy demand and market interest rates.
Unfortunately, most parsimonious theoretical models typically treat these times series as model
outputs rather than model inputs. It remains the delicate task of the modeller to come up with an
analytic apparatus which at is both concise and flexible.

Our paper highlights the issues which arise in reconciling theory and practice for dynamic
general equilibrium analysis. Three illustrative models are considered. The first is a canonical model of
intertemporal choice which illustrates the improvements in model flexibility which can be achieved
with a modest increase in model complexity. The second model is a closed economy Ramsey growth
model with two infinitely lived representative agents. These agents differ with respect to population

growth, labor productivity growth and size. Calibration strategies which seem to work effectively in the
single agent setting are less easily applied with multiple consumers, particularly when hetergeneous
growth rates induce implausible levels of lending or borrowing. The final model is a simple two region
energy-economy model which is calibrated to independent forecasts of regional GDP, regional energy
demand and international energy prices. The model is parameterized with rought values for OECD and
non-OECD regions over a 100 year horizon. The model provides a concrete example of how to work
backwards from baseline equilibrium outcomes to infer consistent technology and preference
parameters.

Thus, this paper addresses fundamental methodological challenges which arise in the calibration
of Ramsey growth models to exogenous assumptions about a baseline growth path. The sample
applications presented here are are primarily motivated by applications for energy and climate policy
analysis, but these techniques could be readily adapted in other fields of economics.

When we build a calibrated equilibrium model, the model is formulated so that it replicates
observed economic activities. In the static setting, this is fairly straightforward, because the benchmark
database is typically based on a social accounting dataset. Subsequent policy analysis is then based on
counter-factual simulations. These simulations explore how the simulated economy respond to
changes in exogenous parameters, e.g. tax rates, technology, terms of trade or primary factor
endowments.

We can formalize this description by first designating the price (p ) and quantity (q ) variables
which define an equilibrium by a vector z which is a concatenation of q and p . The equilibrium
calculation depends on the base year equilibrium values, z̄ , the base year and counterfactual policy
parameters, τ̄ and τ, and the elasticities,σ. The counterfactual equilibrium conditions can be
expressed as:

E (z ;τ,σ; z̄ , τ̄)⊥ z ≥ 0

in which “⊥” designates complementary slackness between the equilibrium conditions and the
equilibrium variables (e.g., the equilibrium could involve zero prices or idle production activities).

The equilibrium conditions define equilibrium variables as an implicit function of the remaining
parameters,z (τ; τ̄, z̄ ,σ). Most model applications involve assessing how changes in τ produce changes
in z ∗, contingent on assumptions and empirical evidence regarding elasticities,σ

Benchmark replication provides a necessary (but not sufficient) verification of a model
specification. When policy parameters assume reference year values, the model should replicate the
reference equilibrium. In terms of the model portrayed as an implicit function, benchmark replication
consists of verifying the following identity:

z (τ̄; τ̄, z̄ ,σ) = z̄

Elasticites,σ, are typically free parameters, and thus the benchmark replication identity should
hold for any assumed elasticity values. The benchmark equilibrium, however, ties down one parameter
apiece, so the model is able to accommodate as many “calibrated parameters” as there are variables in
the benchmark dataset.

The extension of these methods to dynamic settings introduces a number of methodological
challenges. We typically then work from model inputs consisting of a set of base year transactions, and
part of the model input consists of assumptions regarding future development of the economy. The key
input assumption concerns potential GDP, typically embodied in a time series of labor supply
measured in efficiency units. In addition, a baseline simulation may include assumptions about the
future evolution of energy demand, accounting for autonomous energy enhancing improvements
(AEEI).

When a dynamic model is based on a steady-state growth path, the construction of a baseline
growth path is more or less equivalent to working with a static model. In the absence of changes in
demographics or technology, the baseline equilibrium growth path involves proportional scaling of
base year demands, all equilibrium values can be inferred directly from the reference year, growth rate
and interest rate.

When a dynamic model is instead based on a non-stationary baseline, model calibration can be
considerably more difficult. In such models, consumer choices over consumption are driven by
intertemporal choice, and the parameters of intertemporal preferences can be only indirectly
calibrated, on the basis of base year interest rates and assumptions regarding future growth.
Additionally, particular aspects of the baseline growth path may be exogenous inputs. In climate policy
analysis, the time path of energy demands through the model horizon is often taken as exogenous. In a
neoclassical framework, technical change is exogenous, yet it remains a subtle maneuver to adjust
technology parameters so as to maintain consistency between the computed baseline growth path and
the exogenous energy demand profile.

If the benchmark dataset is based on a single year, t = 0, the model projects values from this year
into one or more years in the future, based on an exogenous set of assumptions regarding future
growth, denoted here by a vectorφ. The implicit function which projects from the base year into the
future, consistent with reference policy instruments:

z t = z t (τ; τ̄, z̄ 0,φ,σ)

The baseline growth assumptions include a diverse collection of factors. Demographic growth
assumptions are invariably part of these inputs, but there may be a number of other assumptions, such
as those regarding productivity growth and autonomous changes in energy intensity.

If the model runs over T time periods, and there are n equilibrium variables per period, we then
have n × (T −1)more variables than there are parameters. Calibration of model parameters therefore
requires an implicit or explicit assumptions about the baseline growth path.

Thus, there are two distinct issues involved in setting up the reference equilibrium. First, there is
the issue of baseline growth rates and interest rates, equilibrium values which are directly related to
assumptions about discounting, factor productivity and population growth. Second, one may be
concerned about how technologies and preferences change over time and affect intra-temporal
allocations. In climate policy applications, for example, an important input to the modeling process
involves changes over time in the baseline demand for energy. In trade policy analyses, we need to
account for how heterogeneous GDP growth affects import demand, export supply and inter-regional
terms of trade. All of these considerations are part of the “art” of formulating a dynamic equilibrium
model.

2 The DICE Model

Nordhaus [2008] uses a Ramsey growth model, DICE, to study long-term climate policy issues.

Figure 1: Optimal Policy Comparisons in Dice 1994

3 Calibration with Nonstationary Discounting

The DICE model works with an instantaneous per-capita utility function of the form:

U (c ) =
c 1−a −1

1−a
(1)

in which parameter a is defined as elasticity of the marginal utility of consumption.1

Intertemporal welfare in this Ramsey model is defined on the basis of population and an
exogenous (and constant) discount rate, δ:

W =
∑

t

Pt

(1+δ)t
U (c t ) (2)

When the population growth rate in year t equals n t , an index of population in time t is given by:

Pt =
t−1
∏

τ=0

(1+n t ). (3)

Let Γt represent future-value GDP in year t , a parameter which reflects growth of both
population and labor productivity:2

Γt = Γ̄
t−1
∏

τ=0

(1+γt ) (4)

Per-capita GDP in time t is simply Γt /Pt .

If we convert to aggregate consumption as a decision variable, we define:

C t =Pt c t (5)

We can then define W̃ as a scaled and recentered welfare index, defined in terms of aggregate
consumption:

W̃ =
∑

t

P a
t

(1+δ)t
U (C t ) (6)

Optimal choices are invariate with respect to scaling or shifting of W , hence maximization of W̃
is equivalent to maximization of W . The consumers intertemporal choice problem can be written as:

max W̃

s.t.
∑

t

p t C t =M

in which M represents value of current and future income flow – aggregate societal wealth, e.g.
M =

∑

t p t Γt .

1This parameter corresponds to the inverse of the intertemporal elasticity of substitution, σT . In the limit as σT → 1, a → 1
and we have the conventional Cobb-Douglas utility function:

U (c ) = ln(c )

2The productivity growth rate is roughly γt )−n t .

Optimization of intertemporal welfare involves equating the marginal rate of substitution to the
price ratio, that is:

p t+1

p t
=

∂ W̃ /∂ C t+1

∂ W̃ /∂ C t

=
�

Pt+1

Pt

�a (1+δ)t

(1+δ)t+1

∂U (C t+1)/∂ C t+1

∂U (C t )/∂ C t

=
(1+n t )a

1+δ

�

C t

C t+1

�a

(7)

The interest rate in period t , rt is defined by the ratio of the present value prices:

rt = p t /p t+1−1

Likewise the consumption growth rate in year t is defined as:

g t =C t+1/C t −1

The first order condition (7) then reduces to:

�

(1+ g t )
(1+n t )

�a

=
1+ rt

1+δ
(8)

Nordhaus reconciles population growth, the discount rate and the base year interest rate through
calibration of the intertemporal elasticity of substitution, thus accounting for growth in per-capita
consumption. The economy is initially on a non-stationary growth path as population growth, n t ,
declines and per-capita income rises. Equation (8) can be applied to calibrate the elasticity of the
marginal utility of consumption to base year statistics: the interest rate, consumption growth rate and
population growth rate, i.e.3

a = log

�

1+δ
1+ r0

�

/ log

�

1+n 0

1+ g 0

�

(9)

The resulting demand system can be interpreted as describing either how interest rates must
evolve to assure that per-capita consumption growth follows per-capita GDP, or how consumption
growth departs from GDP should interest rates remain unchanged.

The textbook version of the intertemporal consumer model is based on only three parameters:
the population growth rate n t , the discount rate, δ, and the elasticity of the marginal utility of
consumption, a . The empirical application of the model faces challenges, particularly in cases where
the growth rate of GDP departs from the rate of population growth. In these cases, optimal
intertemporal choice may induce high rates of lending or borrowing, depending on the extent to which
the discount rate departs from the interest rate (the marginal rate of economic substitution). Faced
with the challenge of reconciling the Ramsey model with empirical regularities (e.g., relatively low rates
of international capital flows), Manne and Rutherford [1994] proposed an alternative approach to

3For x close to zero, log(1+x )≈ x , hence:

a ≈
r0−δ

g 0−n 0

This is a positive value when r0 >δ and g 0 > n 0.

multiregional growth models with demographic transition. They modify the Ramsey maximand to
include non-stationary discounting. Intertemporal welfare is then:

W =
∑

t

Pt

∆t
U (c t ) (10)

where the discount factor to year t is:

∆t =
t−1
∏

τ=0

(1+δt ).

The values of δt can then be selected to approximate an arbitrary time path of per-capita
consumption and interest rates. Solving for δ in (8) with g t = γt (consumption grows with GDP), we
have:

δt = (1+ r0)
�

1+n t

1+γt

�a

−1 (11)

Optimal growth in per-capita consumption then matches growth in per-capita GDP along the baseline
growth path with a constant interest rate, r0.4 This approach offers considerable flexibility, as the
resulting model can be calibrated to an arbitrary growth path in per-capita consumption.

4 A Single Consumer Model

As a concrete illustration of the issues, consider an economy in which a base year interest rate is 7% per
annum, the base year GDP growth rate is 8% and the base year population growth rate is 2%. Following
Nordhaus, we assume that the pure rate of social time preference is fixed at 1% per annum, and use (9)
to calibrate the elasticity of the marginal utility of consumption to the growth rate of per-capita
consumption. This implies that a ≈ 2, coincidentally consistent with independent empirical estimates.

The economy is assumed to follow a non-stationary growth path in which population growth
rates decline from 2% to 0% and GDP growth rates declines from 8% to 1%. These model inputs are
displayed in Figure 3(a).

The macroeconomic consequences of these changes in population and per-capita GDP depend
crucially on whether we assume that the economy is open or closed. If this is a small open economy,
the interest rate is exogenous, i.e. r = r0, and the consumption growth path moves independently of
GDP, and the economy exploits international credit markets to move purchases from the present into
the future.

If we model a closed economy, then the intereste rate is endogenous, and the consumption
growth more or less follows potential GDP growth. If consumption is forced to follow GDP exactly, then
based on (7) with g t = γt ), we find that the domestic interest rate declines over time:

r ∗t = (1+δ)
�

1+γt

1+n t

�a

−1

The interest rate assumptions for the open and closed economy are displayed in Figure 3(b).
When r = r0, the interest rate remains high and future consumption becomes relatively cheaper than
current consumption. As a consequence, current consumption is reduced to finance future
consumption, as illustrated in Figures 3(c) and 3(d).

4If we replace r0 by rt in (11), a baseline growth can we constructed with an exogenous exogenous time profile in the market
interest rate.

Capital flows in this model are calculated on the basis of the present value of income less
consumption:

A t+1 = A t +p t (Γt −C t )

The asset profile for the case of a constant discount rate, measured as a multiple of current
consumption, is illustrated in Figure 4(a). When the discount rate is 1% while the international interest
rate is 7%, the optimal accumulation of international debt reaches a level of 50 times current
consumption in year 2060.

5 A Multi-Consumer Model

The previous section focused on the extreme cases of closed and small open economies. In the first
case, consumption growth follows GDP growth and interest rates adjust accordingly. In the second case
the interest rate is fixed by the international credit markets and consumption moves independent of
GDP, depending on variations in the per-capita GDP and discounting.

When we move to a model with two consumers, we have a intermediate case. Interest rates are
determined through interaction of growth and discounting profiles of the economic agents. We can
illustrate the key idea in a single commodity Ramsey model with two consumers. Unlike the previous
section which focused solely on demand, this section considers a fully specified general equilibrium
model. Output in year t is determined by capital and labor:

Yt =φLαt K 1−α
t

in which α corresponds to the capital value share andφ is a scale factor.

Output in year t is allocated to investment and consumption:

Yt = I t +
∑

h

Cht

where Cht is consumption by agent h in year t .

Capital endowments in period t = 0 are exogenous:

K0 =
∑

h

k̄h ,

and subsequent capital stocks are determined by the decisions of profit-maximizing investors. Period t
invesment (I t ) provides capital in period t +1:

K t+1 =λK t + I t

where λ represents the capital survival share (one minus the depreciation rate).

Finally, labor demand in each period equals labor supply with full employment and inelastic
supplies:

L t =
∑

h

`ht

Note to self: Need to present a table of data on which the model calibration is based.

Consider then an instance of this model in which we have two consumers, North and South. We
specify that the North agent has uniform GDP growth of 2% over the infinite horizon with zero

population growth. South has an base year GDP growth rate of 8% and population growth rate of 2%.
These both decline to long-run steady-state over the next fifty years, following the time path presented
in Figure 3(a).

In all models variants we calibrate the base year interest rate to 7%. Unlike the single consumer
case examined in the previous section, in this model the time path of interest rates is determined
through the interaction of competing claims for current and future output.

The transition path in the reference equilibrium here depends on two factors: discount rates and
relative size of the two consumers. First and foremost, the baseline growth path depends on whether
the calibration procedure follows Nordhaus [2007] or Manne and Rutherford [1994]. In Figure 5, results
displayed on the left are based on a model in which the elasticity of the marginal utility of consumption
is calibrated to the base year interest rate. Results on the right are based on a model with
non-stationary discount rate calculated with (11), assuming that the reference interest rate is equal to
the base year value.

When the model is calibrated using (9), holding the long-run discount rate equal to 1%, the
relative size of the North and South agents have a substantial impact on the resulting growth path.
When South is small relative to North (in scenario 02, South is 2% of base year GDP), the model
“mimics” the open economy growth path described above. In this case, the interest rate is stable and
South consumption increases more rapidly than GDP. When South is large relative to North (scenario
64, South is 64% of base year GDP), the model resembles the closed economy case, as consumption
follows GDP closely, the interest rate falls, and the consumption growth rate falls.

In contrast, when intertemporal preferences are calibrated to GDP growth rates in both regions
and a constant interest rate, the interest rates remain relatively stable (5(d)), consumption levels follow
GDP (5(b)) and consumption growth rates closely follow GDP, resulting in only small levels of lending
and borrowing.

Key: Individual time paths displayed correspond to different benchmark shares in aggregate GDP, ranging from 2%
to 64%. In simulations summarized on the left, the elasticity of the marginal utility of consumption is calibrated
to replicate a 7% interest rate in year 2005. In simulations on the right, discount rates vary over time such that
consumption follows GDP when the interest rate is 7%. The model does reproduce a perfectly balanced growth
path due to second-order effects of the economic transition, but these departures from target are clearly second
order.

Appendix A: Basic Growth (Figure 1)

$title Calibration Strategies Illustrated

* A growth model calibrated to a base year interest
* and population growth is used to simulate the transition
* to a long-run steady-state. Population and GDP per capita
* both evolve gradually from a non-stationary path.
*
* The calibration procedure can either take the intertemporal
* discount rate as given and calibrate the intertemporal
* elasticity to match the base year interest and growth
* rates, or it can take the intertemporal elasticity of
* substitution as given and calibrate the discount rate
* to match the transition path. Alternative calibration
* strategies produce significant differences in the
* marginal product of capital through the transition.

$if not set emuc $set emuc 1
$if not set prstp $set prstp 0.01

$if not set g0 $set g0 0.08
$if not set n0 $set n0 0.02

* Transition speed to the terminal growth path:

$if not set speed $set speed 0.05

$set horizon 2150

set t Time periods /2005*%horizon%/,
tp(t) Time periods to plot /2005*2100/
tl(*) /2020,2040,2060,2080,2100/
tlast(t) Last time period /%horizon%/
tfirst(t) First time period /2005/;

$setglobal domain tp
$setglobal labels tl

parameter

emuc Elasticity of the marginal utility of consumption /%emuc%/
prstp Pure rate social time preference /%prstp%/
sigmaT Intertemporal elasticity
r Baseline interest rate / 0.07/
g0 Base year GDP growth rate /%g0%/,
n0 Base year population growth rate /%n0%/,
gterm Terminal GDP growth rate /0.01/,
nterm Terminal population growth rate /0 /,
speed Speed of convergence to steady state path /%speed%/,

g(t) GDP growth rate at time t
n(t) Population growth rate at time t
L(t) Labor supply (efficiency units)
P(t) Population
kinit Initial capital stock
cinit Initial consumption level /1/
kterm Terminal capital stock
gamma Reference growth path
beta Discount factor
dfactor Discount factor
pvprice Present value price index,

pvw(t) Present value of wealth;

* Assume that the growth rate starts at g0
* and falls to 1% over time:

g(t) = g0; n(t) = n0;
loop(t,

g(t+1) = (1-speed) * g(t) + speed * gterm;
n(t+1) = (1-speed) * n(t) + speed * nterm;

);
L(t) = 1; P(t) = 1;
loop(t,

L(t+1) = (1 + g(t)) * L(t);
P(t+1) = (1 + n(t)) * P(t);

);

pvprice(t) = power(1/(1+r), ord(t)-1);
dfactor(t) = power(1/(1+prstp),ord(t)-1);
gamma(t) = 1;
beta(t) = 10 * pvprice(t) * L(t);

sigmaT = log( (1+n0)/(1+g0) ) / log( (1+prstp)/(1+r) );

VARIABLES
UTILITY Maximand;

VARIABLES
C(t) Consumption trillion US dollars,
A(t) Asset level (present value);

EQUATIONS
UTIL Objective function,
BUDGET Intertemporal budget constraint (consumer model)
ASSETS Asset profile;

UTIL.. UTILITY =E= SUM(t, beta(t) * (
( LOG(C(t)/gamma(t)) )$(sigmaT=1) +
( ( (C(t)/gamma(t))**(1-1/sigmaT)-1)/(1-1/sigmaT))$(sigmaT<>1)));

budget.. sum(t, pvprice(t)*C(t)) =e= sum(t,pvw(t));

assets(t).. A(t) =e= A(t-1) + pvw(t) - pvprice(t)*C(t);

* Marginal utility:

* mu(t) = beta(t) * C(t)**(-1/sigmaT) (1/gamma(t))

C.L(t) = 1; C.LO(t) = 0.01;

parameter consum Consumption in successive iterations,
cpath Aggregate consumption,
grate Growth rate of aggregate consumption,
irate Interest rate (marginal product of capital)
alevl Asset level;

model consumer / util, budget, assets /;

* Calibrate marginal rate of technical substitution
* to a value consistent with the base year interest
* rate:

sigmaT = log( (1+n0)/(1+g0) ) / log( (1+prstp)/(1+r) );
emuc = 1/sigmaT;
display emuc;

* Assign a price profile which is consistent with th
* exogenously specified growth rate of GDP:

beta(t) = 10 * dfactor(t) * P(t);
gamma(t) = P(t);
loop(t,

pvprice(t+1) = pvprice(t) * (beta(t+1)/beta(t)) * (gamma(t)/gamma(t+1)) *
( gamma(t+1)/gamma(t) * L(t)/L(t+1) )**(1/sigmaT);

);
pvw(t) = pvprice(t)*cinit*L(t);
solve consumer maximizing UTILITY using NLP;

parameter mu(t) Marginal utility;

mu(t) = beta(t)*(C.L(t)/gamma(t))**(-1/sigmaT)/gamma(t);

irate(t,"Closed Economy")$mu(t+1) = 100 * (mu(t)/mu(t+1) - 1);
grate(t,"Closed Economy") = 100 * (C.L(t+1)/C.L(t)-1);
cpath(t,"Closed Economy") = C.L(t)/(cinit*L(t));
alevl(t,"Closed Economy") = A.L(t)/(pvprice(t)*C.L(t));

* Next, compute the consumption path if the
* interest rate remains constant:

pvprice(t) = power(1/(1+r), ord(t)-1);
pvw(t) = pvprice(t)*cinit*L(t);
solve consumer maximizing UTILITY using NLP;

mu(t) = beta(t)*(C.L(t)/gamma(t))**(-1/sigmaT)/gamma(t);

irate(t,"Open Economy")$mu(t+1) = 100 * (mu(t)/mu(t+1) - 1);
grate(t,"Open Economy") = 100 * (C.L(t+1)/C.L(t)-1);
cpath(t,"Open Economy") = C.L(t)/(cinit*L(t));
alevl(t,"Open Economy") = A.L(t)/(pvprice(t)*C.L(t));

parameter grates Growth rates;
grates(t,"GDP") = 100 * g(t);
grates(t,"Population") = 100 * n(t);

Appendix B: Two-Agent Growth (Figure 2)

$title

* A growth model calibrated to a base year interest
* and population growth is used to simulate the transition
* to a long-run steady-state. Population and GDP per capita
* both evolve gradually from a non-stationary path.
*
* The calibration procedure can either take the intertemporal
* discount rate as given and calibrate the intertemporal
* elasticity to match the base year interest and growth
* rates, or it can take the intertemporal elasticity of
* substitution as given and calibrate the discount rate
* to match the transition path. Alternative calibration
* strategies produce significant differences in the
* marginal product of capital through the transition.

$if not set emuc $set emuc 1
$if not set prstp $set prstp 0.01
$if not set sgdp $set sgdp 0.2

$eval ngdp 1-%sgdp%

$if not set g0 $set g0 0.08
$if not set n0 $set n0 0.02

* Transition speed to the terminal growth path:
$if not set speed $set speed 0.05

$set horizon 2120

set t Time periods /2005*%horizon%/,
tlast(t) Last time period /%horizon%/
tfirst(t) First time period /2005/,
a Agents /N, S/;

parameter
prstp Pure rate social time preference /%prstp%/
kvs Capital value share /0.3/
delta Capital depreciation rate /0.07/
r Baseline interest rate / 0.07/

emuc(a) Elasticity of the marginal utility of consumption
/n 0.5, s %emuc%/

sigmaT(a) Intertemporal elasticity
y0(a) Base year GDP index / N %ngdp%, S %sgdp%/
g0(a) Base year GDP growth rate /N 0.02, S %g0%/,
n0(a) Base year population growth rate /N 0, S %n0%/,
gterm Terminal GDP growth rate /0.02/,
nterm Terminal population growth rate /0 /,
c0(a) Base year consumption level
w0(a) Present value consumption
pl0 Base year wage rate
speed Speed of convergence to steady state path /%speed%/,

g(t,a) GDP growth rate at time t
n(t,a) Population growth rate at time t
L(t,a) Labor supply (efficiency units)
P(t,a) Population
ke0(a) Initial capital endowment

phi(a) Production scale faster
k0(a) Initial capital stock
gamma Reference growth path
beta Discount factor
dfactor Discount factor
pvprice Present value price index,
pvw Present value of wealth;

$if set bmkchk g0(a) = gterm; n0(a) = nterm;

* Assume that the growth rate starts at g0
* and falls to 1% over time:

g(t,a) = g0(a); n(t,a) = n0(a);
loop(t,

g(t+1,a) = max((1-speed) * g(t,a), gterm);
n(t+1,a) = max((1-speed) * n(t,a), nterm);

);

parameter grates Growth rates;
grates(t,"GDP",a) = g(t,a);
grates(t,"Population",a) = n(t,a);

L(t,a) = 1; P(t,a) = 1;
loop(t,

L(t+1,a) = (1 + g(t,a)) * L(t,a);
P(t+1,a) = (1 + n(t,a)) * P(t,a);
grates(t,"GDP/cap",a)$P(t+1,a) = ((L(t+1,a)/P(t+1,a))/(L(t,a)/P(t,a))-1);

);

k0(a) = y0(a) * kvs / (r + delta);
c0(a) = y0(a) - k0(a) * (g0(a)+delta);
pvprice(t) = power(1/(1+r), ord(t)-1);
dfactor(t) = power(1/(1+prstp),ord(t)-1);
phi(a) = y0(a) / (kvs / (r + delta))**kvs;
pl0 = (1-kvs);
gamma(t,a) = 1;

set mdl Models corresponding to different /"0.5","1","2", dice/;

parameter invest Investment in successive iterations,
consum Consumption in successive iterations,
pivotdata Pivot report data
refval Reference growth paths
cpath Aggregate consumption,
ipath Investment
ypath Output
grate Growth rate of aggregate consumption,
irate Interest rate (marginal product of capital);

refval("GDP",t,a,"rate") = L(t+1,a)/L(t,a)-1;
refval("C", t,a,"level") = L(t,a)*c0(a);

$ontext
$model:ramseymge

$sectors:
C(t,a) ! Consumption
Y(t) ! Production
K(t) ! Capital
I(t) ! Investment

$commodities:
PC(t,a) ! Shadow price of consumption
PY(t) ! Price of output
PK(t) ! Purchase price of capital
RK(t) ! Rental price of capital
PL(t) ! Wage rate
PKT ! Terminal capital stock

$consumer:
RA(a)

$auxiliary:
KT ! Terminal capital stock

$prod:Y(t) s:1
O:PY(t) Q:(sum(a,y0(a)))
I:PL(t) Q:(sum(a,y0(a))) P:pl0
I:RK(t) Q:(sum(a,k0(a))) P:(r+delta)

$prod:K(t)
O:PK(t+1) Q:(1-delta)
O:PKT$tlast(t) Q:(1-delta)
O:RK(t) Q:1
I:PK(t) Q:1

$prod:I(t)
O:PK(t+1) Q:1
O:PKT$tlast(t) Q:1
I:PY(t) Q:1

$prod:C(t,a)
O:PC(t,a)
I:PY(t)

$demand:RA(a) s:sigmaT(a)
D:PC(t,a) Q:gamma(t,a) P:(beta(t,a)/gamma(t,a))
E:PL(t) Q:(y0(a)*L(t,a))
E:PK(tfirst) Q:k0(a)
E:PKT Q:(-k0(a)) R:KT

$constraint:KT
sum(tlast(t), K(t)/K(t-1) - (1+gterm)) =e= 0;

$offtext
$sysinclude mpsgeset ramseymge

Y.L(t) = L(t,"N");
K.L(t) = sum(a,k0(a))*L(t,"N");
I.L(t) = sum(a,k0(a)*(g0(a)+delta))*L(t,"N");

PY.L(t) = pvprice(t);
PK.L(t) = (1+r)*pvprice(t);
PKT.L = sum(tlast,PY.L(tlast));
RK.L(t) = (r+delta) * pvprice(t);
PL.L(t) = pvprice(t)*pl0;

KT.L = power(1+g0("N"),card(t));

loop(mdl,

if (sameas(mdl,"DICE"),
sigmaT(a) = log( (1+n0(a))/(1+g0(a)) ) / log( (1+prstp)/(1+r) );
beta(t,a) = 10 * dfactor(t) * P(t,a);
gamma(t,a) = P(t,a);

else
sigmaT(a) = mdl.val;
gamma(t,a) = c0(a) * L(t,a);
beta(t,a) = pvprice(t)*gamma(t,a);

);

$include RAMSEYMGE.GEN
solve ramseymge using mcp;

pivotdata("path_bau",t,a,mdl) = C.L(t,a);
pivotdata("path_bau",t,"Y",mdl) = Y.L(t);
pivotdata("path_bau",t,"K",mdl) = K.L(t);
pivotdata("path_bau",t,"I",mdl) = I.L(t);
pivotdata("grate_bau",t,a,mdl) = C.L(t+1,a)/C.l(t,a)-1;
pivotdata("irate_bau",t,a,mdl)$PY.L(t+1) = PY.L(t)/PY.L(t+1) - 1;

);

refval("GDP",t,a,"rate") = L(t+1,a)/L(t,a)-1;
refval("C", t,a,"level") = L(t,a)*c0(a);
pivotdata("relpath",t,a,mdl) = pivotdata("path_bau",t,a,mdl)/(L(t,a)*c0(a));
pivotdata("path_bau",t,a,"ref") = L(t,a)*c0(a);
pivotdata("irate_bau",t,a,"ref") = r;

